

Use of a Spreadsheet To Simulate Enzyme Kinetics

Michael F. Bruist


Department of Chemistry and Biochemistry, Philadelphia College of Pharmacy and Science, Philadelphia, PA 19104-4495

A spreadsheet with graphics capabilities can be used to illustrate enzyme kinetics and to explain how computer simulation of a reaction mechanism is accomplished. The exercises described below help illuminate the link between chemical phenomena and their mathematical description. I use these exercises in an advanced undergraduate biochemistry course. They can be done in two or three 4-hour computer laboratory sessions. The first session introduces students to spreadsheets; enzyme kinetics is developed in the remaining sessions. These examples were created on Microsoft Excel 5.0. However, they can be generated on any standard spreadsheet.

The spreadsheet simulates a sequential series of chemical equations by numerically solving the associated differential equations using the Gauss–Cauchy method. No approximations such as steady state or rapid equilibrium are made. Instead, the spreadsheet can be used to illustrate the validity of these approximations. The method is not limited to enzyme kinetics; it is applicable to any system of chemical reactions. Atkinson et al. (1) use spreadsheets in their treatment of chemical kinetics, and Coe (2) uses a spreadsheet to solve differential equations that occur in chemical kinetics.

Introduction to Spreadsheets with Radioactive Decay

A simple system to introduce students to the modeling of reactions with a spreadsheet is the radioactive decay of ^{14}C to ^{14}N via beta emission:



This is a first-order, irreversible reaction with rate constant k ($k = 1.21 \times 10^{-4}$ per year for ^{14}C). You should explain to the students that in a first-order process, the rate is proportional to the amount of reactant present. Thus, doubling the amount of ^{14}C doubles the production of ^{14}N and beta rays (e^-). Mathematically, this is expressed:

$$\Delta^{14}\text{C} = -k[^{14}\text{C}]\Delta t \quad (1)$$

where Δt is a given time period during which the amount of ^{14}C changes by $\Delta^{14}\text{C}$. Now, point out that the amount of ^{14}C changes slightly during Δt . If Δt is sufficiently short this change will be insignificant, and $[^{14}\text{C}]$ can be treated as a constant during Δt . We will return to this point later.

(A)

	A	B	C	D	E
1	Decay of ^{14}C to ^{14}N		$^{14}\text{C} \rightarrow ^{14}\text{N} + e^-$		
2					
3	NAME	VALUE	UNITS	DESCRIPTION	
4	T0	0	years	Initial time	
5	Ci	1000	g	Initial ^{14}C	
6	Ni	0	g	Initial ^{14}N	
7	k	0.000124	1/years	decay constant	
8	ΔT	800	years	time interval	
9					
10	Iteration	Time	^{14}C	^{14}N	Total
11		(years)	(g)	(g)	
12	0	=T0	=Ci	=Ni	=C12+D12
13	=A12+1	=B12+ ΔT	=C12*(1-k* ΔT)	=D12+k* ΔT *C12	=C13+D13

(B)




Figure 1. Spreadsheet for the calculation of the time course of ^{14}C decay. (A) A spreadsheet showing the formulas to be entered into the cells. (B) Portion of the spreadsheet showing the values determined from the equations. Ten iterations are shown. A plot of the amount of ^{14}C (diamonds) and ^{14}N (squares) versus time is given on the right.

The goal is to create a spreadsheet that displays how a given amount of ^{14}C and ^{14}N change with time. This is done by applying eq 1 iteratively as follows. The initial amount of ^{14}C is designated $[^{14}\text{C}]_0$. At successive time intervals of Δt , the new ^{14}C amount is:

$$\begin{aligned} [^{14}\text{C}]_1 &= [^{14}\text{C}]_0 - [^{14}\text{C}]_0 k \Delta t = [^{14}\text{C}]_0 (1 - k \Delta t) \\ [^{14}\text{C}]_2 &= [^{14}\text{C}]_1 (1 - k \Delta t) \\ [^{14}\text{C}]_3 &= [^{14}\text{C}]_2 (1 - k \Delta t) \\ [^{14}\text{C}]_n &= [^{14}\text{C}]_{n-1} (1 - k \Delta t) \end{aligned}$$

Such calculations are especially easy to perform with spreadsheets.

While the amount of ^{14}C decreases, the amount of ^{14}N increases. The rate law for the production of ^{14}N is

$$\Delta^{14}\text{N} = k[^{14}\text{C}]\Delta t$$

and the amount of ^{14}N is calculated iteratively as

$$[^{14}\text{N}]_n = [^{14}\text{N}]_{n-1} + [^{14}\text{C}]_{n-1} k \Delta t$$

Again, this is easily calculated on a spreadsheet.

The amounts of ^{14}C and ^{14}N are represented by values in columns of cells in the spreadsheet; each line represents the reaction Δt time later. A simulation for the decay of ^{14}C is given in Figure 1. The top section of the spreadsheet is used for naming and setting parameters (k , Δt , and the initial conditions). Below that are the iterative calculations. The formulas entered into the cells are shown in part A and the output of the iterations is shown in part B. To the right of the iterations is a graph of ^{14}C and ^{14}N versus time. The graph is updated whenever parameters are changed. In explaining the formulas in the spreadsheet to the students, illustrate how cell references (addresses) work by using the copy and paste commands to show how these references are updated when formulas are placed in a new location. Parameters are in cells that are given a name. Cell addresses of names are not changed during copying and pasting.


The spreadsheet includes an additional column that sums the amount of ^{14}C and ^{14}N present at each iteration. This useful accounting tool helps detect errors in formulas. In a properly calculated process, this sum remains constant throughout the time course. When more complex processes are modeled, such bookkeeping measures are crucial.

The students should experiment with the spreadsheet. They can increase the number of iterations by inserting new lines between the current lines of iteration; inserting in this way ensures that graphs automatically plot the added entries. The parameters such as Δt and k (this would change the isotope) should also be altered. Students should note the relationship between k and half-life (determined from the graph).

By changing Δt , students can see how this parameter determines the accuracy and practicality of the spreadsheet. When Δt is too big, the product $k \cdot \Delta t$ becomes large relative to ^{14}C , and errors accumulate or meaningless negative amounts are determined. When Δt is too small, the model is accurate but unwieldy numbers of iterations are needed. The students should enter 2,000 yr, 800 yr, and 50 yr for Δt in the spreadsheet. The number of iterations is then increased until the students can interpolate to find the time at which 500 g of ^{14}C remains. They should obtain 5,080 yr, 5,450 yr, and 5,730 yr as half-life values. The last value is correct to three significant figures, but it took 115 iterations to obtain. $\Delta t =$

800 yr gives a value that is accurate within 5% in only 7 iterations. In general, a Δt value that decreases the highest amount by 10% in one iteration gives reasonable accuracy.

Once the spreadsheet for a simple first-order reaction has been mastered, the students are ready to explore a simple enzyme-catalyzed reaction:



E, S, X, and P are the unbound enzyme, substrate, intermediate, and product respectively. This more complex scheme couples two reactions. The first reaction is reversible and second order in the forward direction; the second reaction resembles the radioactive decay reaction. This more complex scheme is easily handled by the spreadsheet.

The iterative equations for the spreadsheet are obtained as follows. We are now dealing with species in solution and concentrations rather than amounts are used. First consider the free enzyme E, whose concentration is affected by three processes:

1. When the enzyme reacts with a substrate to create the intermediate, enzyme is consumed:

$$\Delta E_{\text{E+S} \rightarrow \text{X}} = (-k_{+1}[\text{E}][\text{S}])\Delta t$$

2. When intermediate breaks down to reform enzyme and substrate, enzyme is produced:

$$\Delta E_{\text{X} \rightarrow \text{S+E}} = (k_{-1}[\text{X}])\Delta t$$

3. When intermediate converts into enzyme and product, free enzyme is also created:

$$\Delta E_{\text{X} \rightarrow \text{P+E}} = (k_{\text{cat}}[\text{X}])\Delta t$$

All three of these processes must be summed to get the net change in enzyme concentration:

$$\begin{aligned} \Delta E &= \Delta E_{\text{E+S} \rightarrow \text{X}} + \Delta E_{\text{X} \rightarrow \text{S+E}} + \Delta E_{\text{X} \rightarrow \text{P+E}} \\ &= (-k_{+1}[\text{E}][\text{S}] + k_{-1}[\text{X}] + k_{\text{cat}}[\text{X}])\Delta t \end{aligned}$$

Similar equations for the change in concentration can be derived for substrate:

$$\Delta S = (-k_{+1}[\text{E}][\text{S}] + k_{-1}[\text{X}])\Delta t$$

intermediate:

$$\Delta X = (k_{+1}[\text{E}][\text{S}] - k_{-1}[\text{X}] - k_{\text{cat}}[\text{X}])\Delta t$$

and product:

$$\Delta P = (k_{\text{cat}}[\text{X}])\Delta t$$

These formulas were used in the spreadsheet described in Figure 2. The enzyme considered is the mythical enzyme Midase, which converts lead into gold. (Presumably, lithium or a combination of hydrogen and helium nuclei is also produced in order to account for the protons and neutrons that have been removed.) Unfortunately, the enzyme has only been alluded to in bygone times.

The spreadsheet for Figure 2 is organized similarly to the radiocarbon spreadsheet, with the top section reserved for setting rate constants and initial conditions and the bottom section performing the iterative calculations. The formulas are seen in part A; the results of the first five iterations are seen in part B. Each iteration has been set to two microseconds. Note that there are two sums used to confirm conservation of materials. (Despite the mythical nature of the enzyme, it still must adhere to conservation laws.) These check that the



Figure 2. Spreadsheet for Midase. (A) A spreadsheet showing the formulas to be entered into the cells. (B) The values produced for the first five iterations and the 30th iteration are shown. (C) A plot of the concentrations vs time for 20 iterations (40 μs) is shown at the bottom. Key: free Midase (long-dashed line), intermediate (solid line), and gold (short-dashed line). Lead concentrations are off scale. (D) An S/V vs V plot of rates obtained from the spreadsheet. The best-fit line gives $K_M = 0.266 \text{ mM}$ and $V_{\text{max}} = 1.01 \times 10^{-4} \text{ mmol/L}/\mu\text{s}$.

(A)

	A	B	C	D	E	F	G	H	
1	Midase								
2	Midase catalyzes the reaction:								
3	Midase + Pb == Midase•Pb == Midase + Au + Li								
4									
5	NAME	VALUE	UNITS	DESCRIPTION					
6	T0	0.000	μs	Start time					
7	S0	1.000	mM	Initial Substrate (Pb)					
8	E0	0.010	mM	Initial free enzyme					
9	I0	0.000	mM	Initial Intermediate					
10	Au0	0.000	mM	Initial Product (Au)					
11	k1For	0.200	1/((mM)(μs))	Forward rate constant for first reaction					
12	k1Rev	0.040	1/(μs)	Reverse rate constant for first reaction					
13	kcat	0.010	1/(μs)	Forward rate constant for second reaction					
14	ΔT	2.000	μs	Time increment					
15									
16	Iteration	Time	Free Lead	Free Midase	Intermediate	Gold	E Sum	S Sum	
17		(μs)	(mmol/L)	(mmol/L)	(mmol/L)	(mmol/L)	(mmol/L)	(mmol/L)	
18	0	=T0	=S0	=E0	=I0	=Au0	=D18+E18	=C18+E18+F18	
19	=A18-1	=B18+ ΔT	=C18-(C18*D18*k1For* ΔT)+(E18*k1Rev* ΔT)	=D18-(C18*D18*k1For* ΔT)+(E18*(k1Rev+kcat)* ΔT)	=E18+(C18*D18*k1For* ΔT)-(E18*(k1Rev+kcat)* ΔT)	=F18+(E18*kcat* ΔT)	=D19+E19	=C19+E19+F19	
49	Rate=						=(F48-F38)/20		

(B)

	A	B	C	D	E	F	G	H	
16	Iteration	Time	Free Lead	Free Midase	Intermediate	Gold	E Sum	S Sum	
17		(μs)	(mmol/L)	(mmol/L)	(mmol/L)	(mmol/L)	(mmol/L)	(mmol/L)	
18	0	0	1.000	0.01000	0.00E+00	0.00E+00	0.0100	1.000	
19	1	2	0.996	0.00600	4.00E-03	0.00E+00	0.0100	1.000	
20	2	4	0.994	0.00401	5.99E-03	8.00E-05	0.0100	1.000	
21	3	6	0.993	0.00301	6.99E-03	2.00E-04	0.0100	1.000	
22	4	8	0.992	0.00252	7.48E-03	3.40E-04	0.0100	1.000	
23	5	10	0.992	0.00227	7.73E-03	4.89E-04	0.0100	1.000	
48	30	60	0.988	0.00202	7.98E-03	4.47E-03	0.0100	1.000	
49	Rate=						7.98E-05		



sum of all forms of enzyme (free enzyme and intermediate) and the sum of all forms of substrate and product (lead, intermediate, and gold) remain constant.

Part C of Figure 2 presents a plot of the concentrations of free Midase, intermediate, and gold against time. Initially there is a rapid buildup of intermediate, accompanied by a corresponding drop in free enzyme. After about 20 μs , the concentration of intermediate remains essentially constant. However, the gold concentration increases steadily. These time periods correspond to the presteady state and the steady state. Students should change the values of the rate constants and see how these affect the onset of the steady state.

The students use the spreadsheet to determine the steady-state rate of gold formation at various lead concentrations. The rate is found by calculating the slope of the gold production from two points after the steady state has been attained—for example, 40 and 60 μs . In the example for Figure 2, using the initial conditions and rate constants listed in part A, gold concentrations are 2.87×10^{-3} and 4.47×10^{-3} mM at 40 and 60 μs , giving a rate of 7.98×10^{-5} mmol/L/ μs . This

calculation should be incorporated into the spreadsheet. Part D of Figure 2 shows a plot of these results as an S/V versus S plot (3). Such a plot uses the following linearized rearrangement of the Michaelis–Menton equation:

$$\frac{S}{V} = \frac{1}{V_{\text{max}}} S + \frac{K_M}{V_{\text{max}}}$$

It is preferred over a double-reciprocal plot, because it weighs errors more evenly. The best-fit line gives

$$V_{\text{max,app}} = 1.01 \times 10^{-4} \text{ mmol/L}/\mu\text{s} \text{ and } K_{M,\text{app}} = 0.266 \text{ mM}$$

The values predicted from the rate constants and the Midase concentration are

$$V_{\text{max}} = E_0 \cdot k_{\text{cat}} = 1.00 \times 10^{-4} \text{ mmol/L}/\mu\text{s}$$

and

$$K_M = \frac{k_{-1} + k_{\text{cat}}}{k_{+1}} = 0.250 \text{ mM}^1$$

Discuss possible sources of error. These include measuring rates before the steady state is established, and changes in concentrations from the initial values.

Once the students understand the model, I have them alter the spreadsheet to include inhibitors. Different class members set up competitive and noncompetitive inhibitors and then compare the behavior of their models by observing rates at various inhibitor and substrate concentrations. I have also had students model more complex reaction schemes such as the one for chymotrypsin (4). Here, the spreadsheet can be used to demonstrate initial burst kinetics. The relationships among the rate constants required for the burst are easily explored.

Acknowledgments

I would like to thank G. Stark for suggesting Midase and P. Kitts for introducing me to the use of spreadsheets to study problems in kinetics.

Note

1. The result is actually better than it initially appears. The Michaelis–Menton equation is derived using the assumption that $S_0 \gg E_0$. A more accurate version of the enzyme-rate equation is:

$$V = \frac{S_0 E_0 k_{\text{cat}}}{K_M + S_0 + E_0}$$

The logic of this version of the equation makes use of the symmetry of the enzyme and substrate in the reaction. $V_{\text{max}} = E_0 k_{\text{cat}}$, and $K_{M,\text{app}} = K_M + E_0$. In the example, the predicted value for $K_{M,\text{app}}$ is 0.260 mM.

Literature Cited

1. *Dynamic Models in Chemistry: A Workbook of Computer Simulations Using Electronic Spreadsheets*; Atkinson, D. E.; Bower, D. C.; McClard, R. W.; Barkley, D. S., Eds.; Simonson: Marina del Rey, CA, 1990; pp 222–227.
2. Coe, D. J. *Chem. Educ.* **1987**, *64*, 496–497.
3. Cornish-Bowden, A.; Wharton, C. W. *Enzyme Kinetics*; IRL: Washington, DC, 1988; p 8.
4. Voet, D.; Voet, J. G. *Biochemistry*, 2nd ed.; Wiley: New York, 1995; p 389.